
URI-based navigation for Phoenix LiveView
Hex Docs
Installation
Add :cartograph to your list of dependencies in mix.exs:
def deps do
  [
    {:cartograph, "~> 0.2"}
  ]
end
Overview
This library provides end-to-end relative query param parsing and URI patching for LiveViews.
Unlike most other SPA-style frameworks, LiveView has very good built-in capabilities for holding a page's state
in the URL and keeping it in sync with the server, but it provides no built-in utilities for computing relative query param updates.
For example, toggling a single value in a list of multi-selected filters or adding the current page number or size
to an existing set of unrelated query params.
The built-in navigation utilities: Phoenix.LiveView.push_patch/2, Phoenix.LiveView.push_navigate/2, Phoenix.LiveView.JS.patch/2, and Phoenix.LiveView.JS.navigate/2 only allow specifying the entire URI on each call, which means the application developer has to roll their own solution for managing these relative param changes.
Cartograph aims to simplify this process and reduce the boilerplate required to hold all of a LiveView's state in the URL without extra client-side JS code.
The Cartograph.LiveViewParams module adds Phoenix.LiveView.handle_event/3 callbacks to the using LiveView that allow relative query patching via the Cartograph.Component.cartograph_patch/1 and Cartograph.Component.cartograph_navigate/2 functions.
These functions compute the relative query param updates from the current URI of the running LiveView and a keyword list of query manipulation operations.
Another use-case of this library is to allow pre-computing the full URI with relative query patching so that the user can copy or bookmark the link from the HTML in the browser. The Cartograph.Component.parse_patch/2 and Cartograph.Component.parse_navigate/2 functions can be used to build the uri for use with a Phoenix.Component.link/1 component's :patch or :navigate attribute.
Performance
Performance should be ideal when using Cartograph.Component.cartograph_patch/1 since this pushes a patch event to the server, which takes advantage of an internal LiveView optimization that calls Phoenix.LiveView.handle_params/3 directly without an additional round-trip.
Cartograph.Component.cartograph_navigate/2 can't take advantage of this optimization, so it incurs the extra round-trip like any other navigate event.
Lastly, use of the Cartograph.Component.parse_patch/2 and Cartograph.Component.parse_navigate/2 functions for pre-rendering the full URI in links can have a negative impact on performance for pages that render thousands of links due to the URI depending on the assigns, which means every link has to be sent to the client on re-render when the URI assign changes. This should be negligible for most pages, so the better UX of copyable links can be safely preferred for pages that don't have thousands of links on them.
Quickstart
The following example shows the basic usage of cartograph's query patching:
defmodule MyApp.ExampleLive do
  use Phoenix.LiveView

  # use `LiveViewParams` after `Phoenix.LiveView`.
  use Cartograph.LiveViewParams

  alias Phoenix.LiveView.JS
  import Cartograph.Component, only: [cartograph_patch: 1]

  @impl true
  def handle_params(%{"userroles" => selected_roles} = _params, _uri, socket) do
    valid_roles = [:admin, :member]

    updated_socket =
      socket
      |> assign(:selected_roles, Enum.filter(selected_roles, &Enum.member?(valid_roles, &1)))
      |> refresh_user_list()

    {:noreply, updated_socket}
  end

  @impl true
  def handle_params(%{} = _params, _uri, socket) do
    updated_socket =
      socket
      |> assign(:selected_roles, [])
      |> refresh_user_list()

    {:noreply, updated_socket}
  end

  defp refresh_user_list(%Phoenix.LiveView.Socket{assigns: %{selected_roles: []}} = socket) do
    stream_async(socket, :user_list, fn ->
      {:ok, Repo.all(User), reset: true}
    end)
  end

  defp refresh_user_list(%Phoenix.LiveView.Socket{assigns: %{selected_roles: roles}} = socket) do
    stream_async(socket, :user_list, fn ->
      res =
        from(User)
        |> where([user: u], u.role in ^roles)
        |> Repo.all()

      {:ok, res, reset: true}
    end)
  end

  @impl true
  def mount(_params, _session, socket) do
    mounted_socket =
      socket
      |> assign_new(:selected_roles, fn -> [] end)
      |> assign_new(:role_choices, fn -> [{:admin, "Admin"}, {:member, "Member"}] end)

    {:ok, mounted_socket}
  end

  def render(assigns) do
    ~H"""
    <section id="role-multi-select">
      <details phx-mounted={JS.ignore_attributes(["open"])}>
        <summary>
          Selected User Roles
        </summary>

        <div>
          <%= for {data_value, display_value} <- @role_choices do %>
            <input
              type="checkbox"
              checked={Enum.member?(@selected_roles, data_value)}
              phx-click={cartograph_patch(query: [toggle: %{"userroles[]" => data_value}])}
            />
            {display_value}
            <br />
          <% end %>
        </div>
      </details>
    </section>
    """
  end
end
This LiveView allows filtering a list of users by zero or more roles toggled with a multi-select UI.
The array of roles to filter on is maintained in the URL via the userroles[] query param key.
The Cartograph.Component.cartograph_patch/1 function used in the template constructs a new URL query from the existing page's URL with the userroles[] query param for the corresponding value toggled.
In other words, if the current URL is /users, and we call cartograph_patch(query: [toggle: %{"userroles[]" => :admin}]), the resulting URL used in the Phoenix.LiveView.push_patch/2 event will be: /users?userroles[]=admin.
Toggling %{"userroles[]" => :member} on the result of the previous call would give /users?userroles[]=admin&userroles[]=member.
Toggling :admin again would give /users?userroles[]=member.
The :query keyword list supports the following operations:
	:set
	:add
	:merge
	:remove
	:toggle

See the documentation for the Cartograph.Component.query_opts/0 type for details of how each operation can be used.
Cartograph Parsers
Cartograph also provides some conveniences for reducing boilerplate when parsing params.
Let's refactor our example LiveView to use a private function for parsing the selected_roles in order to make our handle_params/3 more extensible:
  def parse_selected_roles(socket, %{"userroles" => selected_roles}) do
    valid_roles = [:admin, :member]
    socket
    |> assign(:selected_roles, Enum.filter(selected_roles, &(Enum.member?(valid_roles, &1))))
  end

  def parse_selected_roles(socket, %{}), do: assign(socket, :selected_roles, [])

  @impl true
  def handle_params(params, _uri, socket) do
    updated_socket =
      socket
      |> parse_selected_roles(params)
      |> refresh_user_list()

    {:noreply, updated_socket}
  end
This is a good pracice in general, but we can make this more data-driven with the @cartograph_parser module attribute:
  @cartograph_parser [
    handler: &__MODULE__.parse_params/3,
    keys: [:selected_roles],
  ]

  def parse_params(socket, params, :selected_roles), do: parse_selected_params(socket, params)

  @impl true
  def handle_params(params, _uri, socket), do: {:noreply, refresh_user_list(socket)}
The @cartograph_parser module attribute adds the handler function to the cartograph handle_params/3 lifecycle hook and runs it for each of the elements in keys.
The :handler function has the signature Cartograph.CartographParser.param_handler/0. Handler functions take the socket, the params, and an arbitrary atom for matching implementations.
So in our example above, this is roughly equivalent to the following:
  def handle_params(params, _uri, socket) do
    updated_socket =
      socket
      |> parse_params(params, :selected_roles)
      |> refresh_user_list(socket)

    {:noreply, updated_socket}
  end
If we added another key e.g. :current_role to the @cartograph_parser keys array, then this would add a call to parse_params(socket, params, :current_role) to the reduction in the handle params lifecycle hook.
If the logic of our handle_params/3 callback can be satisfied entirely with :handler functions, then we can further reduce the boilerplate by providing the handle_params: true option to use Cartograph.LiveViewParams:
  use Phoenix.LiveView

  use Cartograph.LiveViewParams, handle_params: true

  alias Phoenix.LiveView.JS
  import Cartograph.Component, only: [cartograph_patch: 1]

  @cartograph_parser [
    handler: &__MODULE__.parse_params/3,
    keys: [:selected_roles],
  ]

  def parse_params(socket, params, :selected_roles), do: parse_selected_params(socket, params)

  @impl true
  def mount(_params, _session, socket) do
    mounted_socket =
      socket
      |> assign_new(:selected_roles, fn -> [] end)
      |> assign_new(:role_choices, fn -> [{:admin, "Admin"}, {:member, "Member"}] end)

    {:ok, mounted_socket}
  end

  def render(assigns) do
    ~H"""
    <section id="role-multi-select">
      <details phx-mounted={JS.ignore_attributes(["open"])}>
        <summary>
          Selected User Roles
        </summary>

        <div>
          <%= for {data_value, display_value} <- @role_choices do %>
            <input
              type="checkbox"
              checked={Enum.member?(@selected_roles, data_value)}
              phx-click={cartograph_patch(query: [toggle: %{"userroles[]" => data_value}])}
            />
            {display_value}
            <br />
          <% end %>
        </div>
      </details>
    </section>
    """
  end
When handle_params: true is provided, the Cartograph.LiveViewParams.__using__/1 macro will add a boilerplate implementation of Phoenix.LiveView.handle_params/3 to the using LiveView.
This works well for common query params such as pagination or dynamic filters when defining a "base" LiveView along with helper modules for the shared parsing functions. For example:
defmodule MyApp.BaseLiveView do
  use Phoenix.LiveView
  use Cartograph.LiveViewParams, handle_params: true
end

defmodule MyApp.QueryHelpers do
  import Phoenix.Component, only: [assign: 3]

  def parse_params(socket, %{"userroles" => selected_roles}, :selected_roles) do
    valid_roles = [:admin, :member]

    socket
    |> assign(:selected_roles, Enum.filter(selected_roles, &Enum.member?(valid_roles, &1)))
  end

  def parse_params(socket, %{}, :selected_roles), do: assign(socket, :selected_roles, [])

  def parse_params(socket, %{"current_role" => current_role}, :current_role)
      when current_role in [:admin, :member] do
    assign(socket, :current_role, current_role)
  end

  def parse_params(socket, %{}, :current_role), do: socket
end

defmodule MyApp.ExampleLiveView do
  use MyApp.BaseLiveView

  alias Phoenix.LiveView.JS
  alias MyApp.QueryHelpers
  import Cartograph.Component, only: [cartograph_patch: 1]

  @cartograph_parser [
    handler: &QueryHelpers.parse_params/3,
    keys: [:selected_roles],
  ]

  @impl true
  def mount(_params, _session, socket) do
    mounted_socket =
      socket
      |> assign_new(:selected_roles, fn -> [] end)
      |> assign_new(:role_choices, fn -> [{:admin, "Admin"}, {:member, "Member"}] end)

    {:ok, mounted_socket}
  end

  def render(assigns) do
    ~H"""
    <section id="role-multi-select">
      <details phx-mounted={JS.ignore_attributes(["open"])}>
        <summary>
          Selected User Roles
        </summary>

        <div>
          <%= for {data_value, display_value} <- @role_choices do %>
            <input
              type="checkbox"
              checked={Enum.member?(@selected_roles, data_value)}
              phx-click={cartograph_patch(query: [toggle: %{"userroles[]" => data_value}])}
            />
            {display_value}
            <br />
          <% end %>
        </div>
      </details>
    </section>
    """
  end
end
See the api docs for the Cartograph.LiveViewParams and Cartograph.Component modules for detailed documentation.

This module provides functions for relative query patching and navigation event dispatching in templates.
Examples
Linkable Breadcrumbs
  use Phoenix.Component
  import Cartograph.Component, only: [parse_patch: 2]

  attr :trail, :list, required: true
  attr :curr_uri, :string, required: true
  attr :id, :string, required: true

  def breadcrumbs(assigns) do
    ~H"""
    <div id={@id}>
      <.link patch={
        parse_patch(@curr_uri, query: [remove: Enum.map(@trail, fn {p, _} -> p end)])
      }>
        Top
      </.link>
      <span>&nbsp;&gt;</span>
      <%= for {{query_param, display_label}, idx} <- Enum.with_index(@trail) do %>
        <span>
          <.link patch={
            parse_patch(@curr_uri,
              query: [
                remove: Enum.slice(@trail, (idx + 1)..-1//1) |> Enum.map(fn {p, _} -> p end)
              ]
            )
          }>
            {display_label}
          </.link>
          <span>&nbsp;&gt;</span>
        </span>
      <% end %>
    </div>
    """
  end
Old-School Pagination Widget
  use Phoenix.Component
  import Cartograph.Component, only: [cartograph_patch: 1]

  defp compute_prev_page(page_no, page_count) do
    if page_no == 1 do
      page_count
    else
      page_no - 1
    end
  end

  defp compute_next_page(page_no, page_count) do
    if page_no == page_count do
      1
    else
      page_no + 1
    end
  end

  attr :page_no, :integer, required: true
  attr :page_count, :integer, required: true
  attr :id, :string, required: true

  def pagination(assigns) do
    ~H"""
    <div id={@id}>
      <div>
        <button phx-click={
          cartograph_patch(
            query: [merge: %{page_no: compute_prev_page(@page_no, @page_count)}]
          )
        }>
          Prev
        </button>
        <p>Page {@page_no} of {@page_count}</p>
        <button phx-click={
          cartograph_patch(
            query: [merge: %{page_no: compute_next_page(@page_no, @page_count)}]
          )
        }>
          Next
        </button>
      </div>
      <div>
        <p>Jump to:</p>
        <input
          type="text"
          pattern="�+"
          value={@page_no}
          phx-keydown={cartograph_patch(query: [merge: %{page_no: :phx_value}])}
          phx-key="Enter"
        />
      </div>
    </div>
    """
  end
Simple Stateful Selectable
  use Phoenix.Component
  import Cartograph.Component, only: [cartograph_patch: 1]

  attr :display_label, :string, required: true
  attr :choices, :list, required: true
  attr :selected, :string, required: true
  attr :query_param, :string, required: true
  attr :id, :string, required: true

  def generic_select(assigns) do
    ~H"""
    <label for={@id}>{@display_label}</label>
    <select id={@id}>
      <option value="" phx-click={cartograph_patch(query: [remove: [@query_param]])}>
        No Selection
      </option>
      <%= for {value, display_text} <- @choices do %>
        <option
          value={value}
          selected={@selected == value}
          phx-click={cartograph_patch(query: [merge: %{@query_param => value}])}
        >
          {display_text}
        </option>
      <% end %>
    </select>
    """
  end
Stateful Column Sort Button
Cycles through no sort > ascending > descending > no sort
  use Phoenix.Component
  import Cartograph.Component, only: [cartograph_patch: 1]

  defp sort_toggle(field_name, :asc = _curr_sort_order) do
    ops = [merge: %{"sort[]" => %{field_name <> "-asc" => field_name <> "-desc"}}]
    cartograph_patch(query: ops)
  end

  defp sort_toggle(field_name, :desc = _curr_sort_order) do
    cartograph_patch(query: [remove: %{"sort[]" => field_name <> "-desc"}])
  end

  defp sort_toggle(field_name, nil = _curr_sort_order) do
    cartograph_patch(query: [add: %{"sort[]" => field_name <> "-asc"}])
  end

  attr :display_text, :string, required: true
  attr :field_name, :string, required: true
  attr :sort_order, :atom, default: nil
  attr :sort_idx, :integer, default: nil

  def sort_button(assigns) do
    ~H"""
    <button class="common-btn" phx-click={sort_toggle(@field_name, @sort_order)}>
      <span>{@display_text}</span>
      <Heroicons.icon :if={@sort_order == :desc} name="arrow-down" class="inline-icon" />
      <Heroicons.icon :if={@sort_order == :asc} name="arrow-up" class="inline-icon" />
      <span if={@sort_order != nil} class="text-sm">{@sort_idx}</span>
    </button>
    """
  end

      


      
        Summary


  
    Types
  


    
      
        query_opts()

      


          A keyword list representing the query patching operations to apply to an existing URI.



    





  
    Functions
  


    
      
        cartograph_navigate(uri, opts \\ [])

      


        Push a live navigation event to the server with the href computed by relative query parsing.



    


    
      
        cartograph_patch(opts \\ [])

      


        Push a live patch event to the server with the href computed by relative query parsing.



    


    
      
        parse_navigate(uri, opts \\ [])

      


        Parses a new URI suitable for use in a live navigation operation from the provided uri and opts.



    


    
      
        parse_patch(uri, opts \\ [])

      


        Parses a new path suitable for use in a live patch operation from the provided uri and opts.



    





      


      
        Types


        


  
    
      
    
    
      query_opts()



        
          
        

    

  


  

      

          @type query_opts() :: Keyword.t()


      


  A keyword list representing the query patching operations to apply to an existing URI.
  Unless otherwise specified in the specific option section, the values of the keyword items should be maps. Keys and values of the maps can be atoms or strings. Values can also be lists of atoms or strings in which case, the resulting query string will have one ocurrence of that key for each value in the list.
  The operations are applied cumulatively in the order the keywords are provided, so the behavior of any arbitrary combination of operations is well-defined.
  Providing a list as a value in the map will set multiple values for that key in the resulting query string.
  For example, given the map %{selected_role: [:admin, :member]}, this would be parsed out to: ?selected_role=admin&selected_role=member in the resulting query string. The exact semantics for how these values get applied depends on the operation being used.
Valid Options
	:set - replaces the whole query relative to the current document with the key-value pairs provided.
	Example with single-values:	base query: ?foo=bar
	operation: query: [set: %{bar: :baz}]
	result: ?bar=baz


	Example with multi-values:	base query: ?foo=bar
	operation: query: [set: %{bar: [:baz, :qux]}]
	result: ?bar=baz&bar=qux




	:add - appends the provided key-value pairs into the query string without checking for existing keys. This can create duplicates, which is needed for array-valued params.
	Example with single-values:	base query: ?foo=bar
	operation: query: [add: %{foo: :baz, bar: :baz}]
	result: ?foo=bar&foo=baz&bar=baz


	Example with multi-values:	base query: ?foo=bar
	operation: query: [add: %{foo: [:baz, :qux], bar: :baz}]
	result: ?foo=bar&foo=baz&foo=qux&bar=baz




	:merge - the same as :add but replaces existing keys instead of appending duplicates.
This option allows specifying a map as the value of a key in the top-level key-value map.
If a regular scalar or list is provided as the value of the key, then all ocurrences of that key are removed and the provided values are added at the end of the query string.
If a map is provided as the value for a key, it is used to match values to replace in-place for the corresponding key, so only params with matching values are removed and order of params is preserved.
	Example with single-values:	base query: ?foo=bar
	operation: query: [merge: %{foo: :baz, bar: :baz}]
	result: ?foo=baz&bar=baz


	Example with multi-values:	base query: ?foo=bar&bar=baz
	operation: query: [merge: %{foo: [:baz, :qux], lorem: :ipsum}]
	result: ?bar=baz&foo=baz&foo=qux&lorem=ipsum


	Example with nested map-values:	base query: ?foo=bar&foo=baz&foo=spam&bar=baz
	operation: query: [merge: %{foo: %{"spam" => :eggs, baz: "qux"}}]
	result: ?foo=bar&foo=qux&foo=eggs&bar=baz




	:remove - The opposite of :add. Removes matching keys.
This option allows passing a list of keys instead of a map of key-value pairs.
If a list of keys is provided, all ocurrences of the matched keys will be removed from the query string regardless of their value.
If a map is provided, only the ocurrences of each key that have a matching value will be removed.
	Example with keys only:	base query: ?foo=bar&foo=baz&bar=baz
	operation: query: [remove: [:foo]]
	result: ?bar=baz


	Example with map single-values:	base query: ?foo=bar&foo=baz&bar=baz
	operation: query: [remove: %{foo: :baz}]
	result: ?foo=bar&bar=baz


	Example with map multi-values:	base query: ?foo=bar&foo=baz&foo=qux&bar=baz
	operation: query: [remove: %{foo: [:baz, :qux]}]
	result: ?foo=bar&bar=baz




	:toggle - toggles the provided key-value pairs into or out of the query string.
This operation has the semantics of :add for any key-value pairs not in the current query
and :remove for any that are in the current query.
	Example with single-values:	base query: ?foo=bar
	operation: query: [toggle: %{foo: :bar, bar: :baz}]
	result: ?bar=baz


	Example with multi-values:	base query: ?foo=bar
	operation: query: [toggle: %{foo: [:bar, :baz], bar: :baz}]
	result: ?foo=baz&bar=baz

Push a live navigation event to the server with the href computed by relative query parsing.
uri can be a URI.t/0 struct or a string:
	cartograph_navigate(@cartograph_uri, query: [merge: %{page_no: 1}])
	cartograph_navigate(~p"/users", query: [merge: %{page_no: 1}])

Options
	:query - the query operations to apply, see: query_opts/0
The special placeholder value :phx_value will be replaced by the current value
of the input sending the event.
	Example:	template: <input type="number" phx-key="Enter" phx-keydown={cartograph_navigate(@current_uri, query: [merge: %{"page_no" => :phx_value}])} />
	current uri: https://localhost:4000/users?page_no=1
	user input: 4
	resulting navigate uri: https://localhost:4000/users?page_no=4




	:loading - passed through to Phoenix.LiveView.JS.push/2 as-is.

	:page_loading - passed through to Phoenix.LiveView.JS.push/2 as-is.

Push a live patch event to the server with the href computed by relative query parsing.
Options
	:query - the query operations to apply, see: query_opts/0
The special placeholder value :phx_value will be replaced by the current value
of the input sending the event.
	Example:	template: <input type="number" phx-key="Enter" phx-keydown={cartograph_patch(query: [merge: %{"page_no" => :phx_value}])} />
	current uri: /users?page_no=1
	user input: 4
	resulting patch uri: /users?page_no=4




	:loading - passed through to Phoenix.LiveView.JS.push/2 as-is.

	:page_loading - passed through to Phoenix.LiveView.JS.push/2 as-is.

Parses a new URI suitable for use in a live navigation operation from the provided uri and opts.
uri can be a URI.t/0 struct or a string:
	parse_navigate(@cartograph_uri, query: [merge: %{page_no: 1}])
	parse_navigate(~p"/users", query: [merge: %{page_no: 1}])

Options
	:query - the query operations to apply, see: query_opts/0
	:phx_value - the value of this option will replace any ocurrences of :phx_value
in the :query operations.	Example:	starting uri: https://localhost:4000/users?page_no=1
	opts: query: [merge: %{"page_no" => :phx_value}], phx_value: 2
	result: https://localhost:4000/users?page_no=2

Parses a new path suitable for use in a live patch operation from the provided uri and opts.
uri can be a URI.t/0 struct or a string:
	parse_patch(@cartograph_uri, query: [merge: %{page_no: 1}])
	parse_patch(~p"/users", query: [merge: %{page_no: 1}])

Options
	:query - the query operations to apply, see: query_opts/0
	:phx_value - the value of this option will replace any ocurrences of :phx_value
in the :query operations.	Example:	starting uri: /users?page_no=1
	opts: query: [merge: %{"page_no" => :phx_value}], phx_value: 2
	result: /users?page_no=2

This module provides integration with the Cartograph navigation events for a LiveView process.
To enable cartograph patch and navigate events, simply use Cartograph.LiveViewParams from
a LiveView module.
This will also add a @cartograph_uri key to the assigns. This key will always contain the current URI from the latest call to Phoenix.LiveView.handle_params/3.
The LiveViewParams module also provides some conveniences for reducing boilerplate related to Phoenix.LiveView.handle_params/3.
See the documentation for __using__/1 for available options.
See the api documentation for Cartograph.Component for how to send patch and navigation events to the using LiveView.

Adds cartograph event handling to a LiveView.
use Cartograph.LiveViewParams,
  cartograph_parser: true,
  handle_params: false,
Options
	:cartograph_parser - if true (default: true) then run any parsing handlers defined with
the @cartograph_parser module attribute on the using LiveView.

	:handle_params - if true (default: false) then add a default implementation of
Phoenix.LiveView.handle_params/3. Useful when using :cartograph_parser to
parse the params.


Important
The LiveViewParams must be used AFTER the LiveView behavior or the on_mount/1 hook
that sets up the param parsing will not be registered. This is needed even if
:cartograph_parser is false because cartograph adds the @cartograph_uri assign to allow
relative query patching.
Good:
use Phoenix.LiveView
use Cartograph.LiveViewParams
Bad:
use Cartograph.LiveViewParams
use Phoenix.LiveView

